Includes easy-entry keystroke listings.

HEWLETT-PACKARD

HP-75

USERS' LIBRARY SOLUTIONS

Electronics

NOTICE

TABLE OF CONTENTS

PROGRAM DESCRIPTION

TRANSISTOR AMP EVALUATION

Given supply voltage (Vcc), the five resistor values R(1)-R(5), and transistor Beta, operating parameters are computed from the formulas:

(See schematic for resistor labels)

Temporary values

```
 R=   R(1) * R(2) / (R(1) + R(2))
VØ=   V * (R(1) / (R(1) + R(2)))
 K=   VØ - * .7
 D=   .026 / A(3) + 2
 T=   B * (R(4) + D)
```

V=Vcc B=Beta

R(1) thru R(5) input as values from schematic

Displayed values

I_B = A(1)= K / (B * (R(4) + R(5)) + R)

V_B = A(2)= VØ - A(1) * R

I_C = A(3)= B * A(1)

V_C = A(4)= V - A(3) * R(3)

V_E = A(5)= A(1) + A(3) * (R(4) + R(5))

Gain = A(6)= R(3) / (R4) + D)

Rin = A(7)= R * T / (R+T)

Answers are rounded to two digits and displayed with proper units and unit multiplier.

SAMPLE PROBLEM

Using the schematic and sample problem values, find I_B, V_B, I_C, V_C, V_E, Gain and input resistance for a Beta of 30, then for a Beta of 300.

Answers: Gain varies 18.96 - 19.2
Rin varies 946 - 2050
Collector current varies by 1.5 m Amps



Sample Problem Values

R1 = 8200
R2 = 3300
R3 = 1000
R4 = 47
R5 = 270
Vcc = 12
First $Beta_1$ 30
Second $Beta_2$ 300

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
	Run program	* TRANSISTOR AMP EVALUATION *	
1	Enter supply volts	Vcc =	12 [RTN]
2	Enter supply R1	R1 =	8200 [RTN]
3	Enter supply R2	R2 =	3300 [RTN]
4	Enter supply R3	R3 =	1000 [RTN]
5	Enter supply R4	R4 =	47 [RTN]
6	Enter supply R5	R5 =	270 [RTN]
7	Enter supply Beta	Beta =	30 [RTN]
8	Answers	I_b = 231.26 μ AMPS	[RTN[
9		V_b = 2.9 VOLTS	[RTN]
10		I_c = 6.94 m AMPS	[RTN]

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
11		V = 5.06 VOLTS	[RTN]
12		V = 2.2 VOLTS	[RTN]
13		Gain = 18.96	[RTN]
14		Rin = 946.14 OHMS	[RTN]
15	Press [RTN] to default to 'B'	New Beta, Run again, or End? B	[RTN]
16	Enter new Beta	Beta =	300 [RTN]
		I_b = 28.15 AMPS	[RTN]
		V_b = 3.38 VOLTS	[RTN]
		I_c = 8.45 m AMPS	[RTN]
		V_c = 3.55 VOLTS	[RTN]
		V_e = 2.68 VOLTS	[RTN]
		Gain = 19.2	[RTN]
		Rin = 2.05K OHMS	[RTN]
17	End program	New Beta, Run again, or End? B	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	* TRANSISTOR AMP EVALUATION *	
2	Enter values:		
2a	Enter Vcc in volts	Vcc =	Vcc [RTN]
2b	Enter R1 in ohms	R1 =	R1 [RTN]
2c	Enter R2 in ohms	R2 =	R2 [RTN]
2d	Enter R3 in ohms	R3 =	R3 [RTN]
2e	Enter R4 in ohms	R4 =	R4 [RTN]
2f	Enter R5 in ohms	R5 =	R5 [RTN]
2g	Enter Beta	Beta =	Beta [RTN]
3	Display answers:	I = value AMPS	[RTN]
		V = value VOLTS	[RTN]/[BACK]
		I = value AMPS	[RTN]/[BACK]
		V = value VOLTS	[RTN]/[BACK]
		V = value VOLTS	[RTN]/[BACK]
		GAIN = value	[RTN]/[BACK]
		Rin = value OHMS	[RTN]/[BACK]
4	Program options:	New Beta, Run again, or End?	B, R, or E [RTN]
	If 'B' then goto 2g		
	If 'R' then goto 1		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A(x)	Computed answers A(1)=I_B, A(2)-V_B, A(3)=I_C, A(4)=V_C, A(5)=V_e, A(6)=Gain A(7)=Rin	V	Supply voltage V
		VØ	Intermediate value
B(x)	Same as A(x) but adjusted for unit multiplier F$	X	For loop counter
R(x)	Input resistance values R1-R5	A$	Labels for prompts
A	Always A=VAL(X$) in non-error input subroutine	B$	Units for answers
B	Beta	F$	Unit multiplier from FNF(F)
D	Intermediate value saved for further computation	K$	Single key input
F	Value passed to function FNF(F) and FNR(F,J)	Q$	Query option input
J	Value passed to function FNR(F,J)	X$	String input always changed to A=VAL(X$) and negative values not allowed.
K,T	Intermediate values		

NOTES AND REFERENCES

1. When viewing answers [RTN] key steps forward to next answer and [BACK] key steps to previous answer.

2. Resistance values of zero ohms are allowed for R4 and R5 only. All inputs in common units and must be positive.

3. Program designed to be most useful as computing parameters based on fixed resistor values and changing Beta.

4. Invalid inputs are ignored and user is re-prompted until valid input is received.

5. Refer to schematic for transistor configuration and resistor labels.

6. From HP Users' Library #3970D.

PROGRAM LISTING

```
10 ! TRANSISTOR AMP EVALUATION
20 ! Computes transistor amp
30 ! operating parameters as a
40 ! function of transistor
50 ! Beta.
60 ! Revision 11/01/82
70 !
80 DELAY 2
90 DIM A$[28],B$[36]
100 DIM R(5),A(7),B(7)
110 ! DEFINE FUNCTIONS
120 ! Fix answer to common unit
130 DEF FNF(F)
140 F$="MK m"&CHR$(12)&"np"
150 J=3
160 IF F<1000 AND F>=1 THEN 240
170 IF F<1 THEN 210
180 IF F=1 THEN 240
190 F=F/1000 @ J=J-1
200 GOTO 160
210 IF J=7 THEN 240
220 F=F*1000 @ J=J+1
230 GOTO 160
240 F$=F$[J,J]
250 FNF=F
260 END DEF
270 !
280 ! Single, upper case key in
290 DEF FNK$
300 K$=KEY$ @ IF K$="" THEN 300
310 FNK$=UPRC$(K$)
320 END DEF
330 !
340 ! Round 'F' to 'J' digits
350 DEF FNR(F,J)
360 F=INT(F*10^J+.5)/10^J
370 FNR=F
380 END DEF
390 !
400 ! Labels
410 A$="  Ib  Vb  Ic  Vc  VeGAIN Rin"
420 ! Units
430 B$=" AMPSVOLTS AMPSVOLTSVOLTS      O
    HMS "
440 DISP "  * TRANSISTOR AMP EVALUATION
     *"
450 P$="Vcc = " @ GOSUB 820 @ V=A
460 FOR X=1 TO 5
470 P$="R"&STR$(X)&" = "
480 GOSUB 830
490 R(X)=A
```

-FNK$ returns single uppercase input

-FNR(F,J) rounds F to J digits

-FNF(F) returns F between 1 and 100 with unit F$

-Prompts and labels

-Units for answers

-Sign on message

-Get supply voltage Vcc

-Get five resistor values R(1) through R(5)

PROGRAM LISTING

```
500 IF R(X) THEN 520
510 IF X=4 OR X=5 THEN 520 ELSE 480          -Only R4 and R5 can be zero
520 NEXT X
530 ! Loop here for new Beta
540 P$="Beta = " @ GOSUB 820 @ B=A           -Get transistor beta
550 ! Compute results                        -Compute all parameters
560 R=R(1)*R(2)/(R(1)+R(2))
570 V0=V*(R(2)/(R(1)+R(2)))
580 K=V0-.7
590 A(1)=K/(B*(R(4)+R(5))+R)
600 A(3)=B*A(1)
610 A(5)=A(1)+A(3)*(R(4)+R(5))
620 A(2)=V0-A(1)*R
630 A(4)=V-A(3)*R(3)
640 D=.026/A(3)+2
650 A(6)=R(3)/(R(4)+D)
660 T=B*(R(4)+D)
670 A(7)=R*T/(R+T)
680 ! Show results
690 FOR X=1 TO 7
700 B(X)=FNF(A(X))                           -B(X) = fixed value of A(X)
710 DISP A$[4*X-3,4*X]&"= ";FNR(B(X),2)
    ;F$&" "&B$[X*5-4,X*5]
720 K$=FNK$
730 IF NUM(K$)#13 AND NUM(K$)#8 THEN 72      -Wait for RTN or BACK keys
    0
740 IF NUM(K$)=13 THEN 760                   -Step forward on RTN
750 X=X-2 @ IF X<0 THEN X=0                  -Step back on BACK but no
                                              further than 0
760 NEXT X
770 DISP "New ";CHR$(194);"eta, ";CHR$(
    210);"un again, or ";CHR$(197);
780 INPUT "nd? ","B"; Q$ @ Q$=UPRC$(Q$)      -Continuation prompt and input
790 IF Q$="R" OR Q$="B" OR Q$="E" THEN       -Wait for valid input
    800 ELSE GOTO 770
800 IF Q$="R" THEN 450
810 IF Q$="B" THEN 540 ELSE 890
820 ! Non-error input                        -Non-error input prompt from P$
                                              returning A
830 DISP P$;
840 INPUT "";X$
850 ON ERROR GOTO 830
860 A=VAL(X$) @ OFF ERROR
870 IF A<0 THEN 830                          -Require positive input
880 RETURN
890 DISP @ DELAY 1 @ STOP                    -Program stops to preserve
                                              variables for inspection
```

PROGRAM DESCRIPTION

COMMON COMPONENTS FOR 555 & 567 IC'S

Trial and error is usually used to find values for common components for use with a 555 oscillator and 567 tone decoder which yield an acceptable tolerance about a given frequency.

This program completes the trial-and-error method by stepping through all common values of components and testing each combination to see if it is within specified tolerance.

Every value in array A(I) is tested by one of the two formulas:

555 = C = 1.433 / (R(A) + R(B) + R(B)) * F

567 = C = 1/(R(A) * F)

Component values are displayed along with the <u>computed</u> operating frequency.

Upper tolerance is held in variable 'U'.

Lower tolerance is held in variable 'L'.

i.e.: L = .95 = -5%
U = 1.05 = +5% ± 5% tolerance

SAMPLE PROBLEM

A) What values of R and C should be used to decode 1200 Hz ±5% in a 567 tone decoder?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
A_1	Run program, select 567	Select: 555 or 567? 555	567 [RTN]
2	Sign-on message	*** 567 TONE DECODER ***	
3	Enter frequency	Frequency in Hz =	1200 [RTN]
4	Step and test	Searching for first combination	
5	Show components	R(A)=180Ω C=4.7μfd	[RTN]
6	Now show computed frequency	Frequency is 1.82KHz	[RTN]
7	End program	<u>V</u>iew again, <u>M</u>ore, or <u>E</u>nd?	E [RTN]

SAMPLE PROBLEM

B) What values of R(A), R(B) and C are suitable for the operation of a 3.3 KHz oscillator?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
B 1	Run program select 555	Select: 555 or 567? 555	[RTN]
2	Sign-on message	*** 555 OSCILLATOR ***	
3	Enter frequency	Frequency in Hz =	3300 [RTN]
4	Step and test	Searching for first combination	
5	Show component values	R(A)=100Ω R(B)=180Ω	[RTN]
6	Sow capacitor	C=1µfd	[RTN]
7	Show computed frequency	Frequency is 3.137 KHz	[RTN]
8	End program	View again, More, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program, select operation	Select: 555 or 567? 555	555 or 567 [RTN]
2	If 567 then goto 14		
3	555 selected	**555 OSCILLATOR**	
4	Input frequency	Frequency in Hz =	Freq. [RTN]
5	Searching array	Searching for first combination	
6	Display resistor values	R(A)=100Ω R(B)=220Ω	[RTN]
7	Display capacitor value	C=2.2	[RTN]
8	Show actual frequency	Frequency is 1.215 KHz	[RTN]
9	Continuation options	View again, More, or End?	V, M, E [RTN]
10	If View then 6		
11	If End then 24		
12	More - continue search		
13	Goto 6	Searching for more combinations	
14	567 selected	**567 TONE DECODER**	
15	Input frequency	Frequency in Hz =	Freq. [RTN]
16	Search for combination	Searching for first combination	
17	Display results	R(A)=100 C=5	[RTN]
18	Display frequency	Frequency is 2KHz	[RTN]
19	Continuation options	View again, More, or End?	V, M, E [RTN]
20	If V then 17		
21	If E then 24		
22	More - continue search	Searching for more combinations	
23	Goto 16		
24	Stop		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A	Array pointer and loop counter	N	Value passed to function
B	Loop counter	P	Function pointer
C	Capacitor value	Q	Operation flag 1=555 Ø=567
D	Temporary capacitor value	U	Upper tolerance
F	Frequency selected	Z	Computed frequency
H	Range exponent for C	R1	R(A) from formula
I	Loop counter	R2	R(B) from formula
J	Loop counter	F$	Temporary string
L	Lower tolerance	K$	Single key input
		Q$	Option string input

NOTES AND REFERENCES

Schematic for component values:





PROGRAM LISTING

```
10 ! 567 TONE DECODER
20 ! and 555 OSCILATOR
30 ! Find the common resistor
40 ! and capicator combinations
50 ! for either of the
60 ! two circuits.
70 ! Revision 11/01/82
80 DELAY 2
90 DIM A(25),R(77)
100 ! Define Functions
110 !
120 ! Wait for 'RTN' key
130 DEF FNK$
140 K$=KEY$ @ IF K$="" THEN 140
150 IF NUM(K$)#13 THEN 140
160 FNK$=K$
170 END DEF
180 !
190 ! Round 'N' to 'P' digits
200 DEF FNR(N,P)
210 N=INT(N*10^P+.5)/10^P
220 FNR=N
230 END DEF
240 !
250 ! Fix ohms to common units.
260 DEF FNF(N)                          -FNF(N) sets N = smallest
                                         value>1, sets multiplier
270 F$="MK "
280 P=3
290 IF N<1000 AND N>=1 THEN 320
300 N=N/1000 @ P=P-1
310 IF P#1 THEN 290
320 IF P=3 THEN F$="" ELSE F$=F$[P,P]
330 FNF=N
340 END DEF
350 !
360 ! Which chip? 1=555   0=567
370 INPUT "Select:   555  or  567 ?","5
    55";Q$
380 IF Q$#"555" AND Q$#"567" THEN 370
390 IF Q$="567" THEN Q=0 ELSE Q=1       -Q=1 for 555  and Q=0 for 567
400 ! Main program loop
410 IF Q THEN DISP "  * * * 555 OSCILLA
    TOR * * *" ELSE DISP "   * * * 567
    TONE DECODER * * *"
420 R1,C=0 @ IF Q THEN R2=0
430 ! Set upper & lower tolerance
440 U=1.05 @ L=.95                      -Tolerance set to 5%
450 ! Non-error input
460 INPUT "Frequency in Hz = "; F$ @ IF -Wait for valid frequency
     F$="" THEN 460
470 ON ERROR GOTO 460
480 F=ABS(VAL(F$)) @ OFF ERROR
490 DELAY 0
```

PROGRAM LISTING

```
500 DISP "Searching for first combinati
    on."
510 ! Search for values
520 FOR I=1 TO 13
530 READ A(I)                                   -Load data values into array A
540 NEXT I
550 ! Common values
560 DATA 5,10,1,2.2,3.3,4.7,1.2,1.5,1.8
    ,3.9,5.6,6.2,8.2
570 ! Load resistor array
580 A=1
590 FOR J=2 TO 7
600 FOR I=3 TO 13
610 R(A)=A(I)*10^J                              -Load values with exponent
                                                 multiplier into array R
620 A=A+1
630 NEXT I
640 NEXT J
650 ! Possible combinations
660 FOR A=1 TO 66
670 IF NOT Q THEN 690                           -Skip search for R(B) if in 567
                                                 mode
680 FOR B=1 TO 66
690 IF Q THEN C=1.443/((R(A)+R(B)+R(B))
    *F) ELSE C=1/(R(A)*F)
700 IF C<10^(-9) THEN 840
710 ! Range C
720 FOR H=1 TO 9
730 D=C/10^(-H)
740 IF 1>D THEN 760
750 IF D<10 THEN 780
760 NEXT H
770 ! Check tolerance
780 FOR I=1 TO 6
790 IF L*A(I)>D THEN 810
800 IF D<U*A(I) THEN 870                        -Display values if they fall
                                                 within tolerance
810 NEXT I
820 IF NOT Q THEN 840
830 NEXT B
840 NEXT A
850 GOTO 1050
860 ! Display results
870 DISP "R(A)=";FNF(R(A));F$&CHR$(17);
    "     ";
880 IF Q THEN DISP "R(B)=";FNF(R(B));F$
    &CHR$(17)
890 IF Q THEN K$=FNK$
900 DISP "  C=";A(I)*10^(6-H);CHR$(12)&
    "fd    "
910 K$=FNK$
920 DISP "Frequency is ";                       -Compute frequency as a
                                                 function of component value
930 IF NOT Q THEN Z=1/(R(A)*A(I)*10^(-H
    ))
```

PROGRAM LISTING

```
 940 IF Q THEN Z=1.443/((R(A)+R(B)+R(B))
     *A(I)*10^(-H))
 950 Z=FNR(FNF(Z),3) @ DISP Z;F$&"Hz"
 960 K$=FNK$
 970 ! Continuation options
 980 DISP CHR$(214);"iew again, ";CHR$(2
     05);"ore, or ";CHR$(197);
 990 INPUT "nd? ","M"; Q$ @ Q$=UPRC$(Q$)
1000 IF Q$="M" THEN 1030
1010 IF Q$="V" THEN 870
1020 IF Q$#"E" THEN 980 ELSE 1050
1030 DISP "Searching for more combinatio
     ns"
1040 GOTO 840
1050 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

OHMS LAW WITH dBm CONVERSION

dBm Section: Given watts computes dBm by the formula dBm=30+10*LOG$_{10}$ (watts). Given dBm computes watts by the formula watts=10^((dBm-30)/10).

OHMS Law Section: Given any two of four inputs, computes volts, Ohms, amps and watts.

Variables Are: Volts = E Current = I = R Watts = P

Formulas Are:

Given E & R	$I = E/R$	$P = E^2/R$
Given E & I	$R = E/I$	$P = E*I$
Given R & I	$E = I*R$	$P = I^2*R$
Given P & E	$R = E^2/P$	$I = P/E$
Given P & R	$E = \sqrt{P*R}$	$I = \sqrt{P/R}$
Given P & I	$E = P/I$	$R = P/I^2$

Answers are displayed one at a time and scanned forward by pressing the [RTN] key or backward by pressing the [BACK] key.

SAMPLE PROBLEM

A) How many watts are represented by -13.5 dBm?

Answer: -13.5 dBm is 44.67 µwatts.

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
A	Run program		
1	Select dBm function	Select: O=Ohms law or D=dBm ?	D [RTN]
2	Sign-on message	dBm TO WATT(s) to dBm CONVERSION	
3	Watts prompt (skip)	Watt(s) =	[RTN]
4	dBm prompt	dBm =	-13.5 [RTN]
5	Answer displayed	-13.5 dBm IS 44.67 µWATTS	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

SAMPLE PROBLEM

B) What power rating is required for a resistor of 220 Ohms across a 12 volt power supply?

Answer: 654.55 m watt - use a 1 watt resistor

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
B	Run program		
1	Select Ohms function	Select: O=Ohms law or D=dBm	O [RTN]
2	Sign on message	** OHMS LAW CALCULATOR **	
3	Input volts	VOLTS =	12 [RTN]
4	Input Ohms	OHMS =	220 [RTN]
5	Answer: volts:	12 VOLTS	[RTN]
	Ohms:	220 OHMS	[RTN]
	amps:	54.55 m AMPS	[RTN]
	(Use 1 watt resistor) watts:	654.55 m WATTS	[RTN]
6	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program		
2	Select Ohms or dBm function	Select: O=Ohms law or D=dBm ?	O or D [RTN]
	D=dBm selected	dBm to WATTS to dBm CONVERSION	
2	Enter value of [RTN]	WATT(s) =	W or [RTN]
3	If value for watts then goto 5		
4	Enter dBm value if known	dBm =	D [RTN] or [RTN]
5	Answer stays until [RTN]	D dBm is W WATTS	[RTN]
6	Programs options:	Run again, View again, or End?	R, V, or E [RTN]
7	If 'V' then goto 5		
8	If 'R' then goto 2 else 14		
	(O=Ohms selected)	** OHMS LAW CALCULATOR **	
9	These four prompts will loop	VOLTS =	value [RTN] or [RTN]
	through until two are answered	OHMS =	value [RTN] or [RTN]
	with values. ([RTN] skips to	AMPS =	value [RTN] or [RTN]
	next prompt)	WATTS =	value [RTN] or [RTN]
10	After two entries the answers	VOLTS = answer	[RTN]/[BACK]
	are displayed: [RTN] steps	OHMS = answer	[RTN]/[BACK]
	to next answer, [BACK] steps	AMPS = answer	[RTN]/[BACK]
	back to previous answer	WATTS = answer	[RTN]/[BACK]
11	Program options:	Run again, View again, or End?	R, V, or E [RTN]
12	If 'V' then goto 10		
13	If 'R' then goto 9		
14	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
C	Input counter (Z-0)	V(4)	Watts (for OHMS law)
D	dBm	W	Watts (for dBm conversion)
F	Variable passed to function FNF(F)	X	For loop counter
J	Variable passed to function FNR(F,J)	A$	Labels and units
N	Binary computed pointer	F$	Units prefix "MK mμnp"
V(1)	Volts	K$	Single key input
V(2)	OHMS	X$	Value input always converted to numeric
V(3)	Amps	Q$	Input needs to be followed by [RTN]

NOTES AND REFERENCES

When viewing OHMS law answers [RTN] steps to next answer, [BACK] steps to previous answer. [RTN] after last (WATTS =) answer will fall into continuation prompt for program options.

PROGRAM LISTING

```
 10 ! OHMS LAW & dBm CONVERSION
 20 ! Computes VOLTS, AMPS, OHMS &
 30 ! WATTS using Ohms Law and
 40 ! given any two known inputs
 50 ! dBm converts WATTS to dBM &
 60 ! dBm to WATTS.
 70 ! Revision 11/01/82
 80 !
 90 DELAY 2
100 DIM A$[24]
110 ! Define Functions
120 ! Fix answer to common unit          -FNF(F) CHANGES F TO SMALLEST
                                          VALUE>1 AND
130 DEF FNF(F)                           -ASSIGNS A UNIT PREFIX F$ OF
                                          M,K,' ',m,u,n, or p
140 F$="MK m"&CHR$(12)&"np"
150 J=3
160 IF F<1000 AND F>=1 THEN 240
170 IF F<1 THEN 210
180 IF F=1 THEN 240
190 F=F/1000 @ J=J-1
200 GOTO 160
210 IF J=7 THEN 240
220 F=F*1000 @ J=J+1
230 GOTO 160
240 F$=F$[J,J]
250 FNF=F
260 END DEF
270 !
280 ! Single, upper case key in          -FNK$ RETURNS SINGLE UPPERCASE
                                          KEYBOARD INPUT
290 DEF FNK$
300 K$=KEY$ @ IF K$="" THEN 300
310 FNK$=UPRC$(K$)
320 END DEF
330 !
340 ! Round 'F' to 'J' digits            -FNR(F,J) Round of F to J
                                          digits
350 DEF FNR(F,J)
360 F=INT(F*10^J+.5)/10^J
370 FNR=F
380 END DEF
390 !
400 ! Prompt and label strings
410 A$=" VOLTS OHMS  AMPS  WATTS"
420 ! Selection Ohms or dBm
430 DISP "Select: ";CHR$(207);"=Ohms la  -Expected entries are
    w or ";CHR$(196);"=dBm ";             underlined when prompted
440 INPUT "?"; K$ @ K$=UPRC$(K$[1,1])
450 IF K$="O" THEN 670                   -Goto Ohms law routine
460 IF K$#"D" THEN 430                   -Try again if not 'dBm'
470 ! Fall into dBm routine.
480 DISP "dBm TO WATT(s) TO dBm CONVERS
    ION"
```

PROGRAM LISTING

```
490 INPUT "WATT(s) = "; X$ @ IF X$="" T    -Skip to dBm entry if no watts
    HEN 540
500 ON ERROR GOTO 490
510 W=VAL(X$) @ OFF ERROR @ IF W<=0 THE    -Wait for positive non-zero
    N 490                                   value for watts
520 D=30+10*LOG10(W)                       -Compute dBm given watts
530 GOTO 580                               -Goto answer display
540 INPUT "dBm = "; X$ @ X$=UPRC$(X$) @
     IF X$="" THEN 1260
550 ON ERROR GOTO 540
560 D=VAL(X$) @ OFF ERROR                  -Wait for any numeric value for
                                            dBm
570 W=10^((D-30)/10)                       -Compute watts given dBm
580 D=FNR(D,2) @ W=FNR(FNF(W),2)           -Round of dBm and watts to two
                                            places
590 DISP D;"dBm  IS ";W;F$&"WATT";         -Display single line answer
600 IF W=1 THEN DISP ELSE DISP "S"
610 K$=FNK$                                -Wait for any key press
620 GOSUB 1230                             -Prompt continue menu and
                                            return entry as Q$
630 IF Q$="V" THEN 590                     -Go back and view again
640 IF Q$="R" THEN 490 ELSE 1260           -Go run again or else stop
                                            program
650 !
660 ! Begin OHMS LAW routine.
670 DISP "  * * OHMS LAW CALCULATOR * *
    "
680 ! Data entry loop
690 ! Set C and clear variables
700 C=2 @ FOR X=1 TO 4 @ V(X)=0 @ NEXT
    X
710 FOR X=1 TO 4                           -Prompt for 4 entries and loop
                                            until 2 are entered
720 DISP A$[X*6-5,X*6]&" = ";
730 IF V(X) THEN DISP V(X);
740 B=SGN(V(X))
750 INPUT ""; X$ @ IF X$="" THEN 820
760 IF X$="Q" THEN 1260
770 ON ERROR GOTO 790
780 V(X)=VAL(X$) @ OFF ERROR @ GOTO 800
790 DISP "OOPS.. "; @ GOTO 720
800 C=C+(B-SGN(V(X)))
810 IF NOT C THEN X=4
820 NEXT X
830 IF C THEN 710                          -When C=0 two values are in
                                            array V()
840 ! Compute answers
850 GOSUB 940                              -Find formulas and compute
                                            answers
860 ! Display the results
870 GOSUB 1110                             -Display the results
880 ! Loop back for more inputs
890 GOTO 680                               -Loop back for more variables
900 !
```

PROGRAM LISTING

```
 910 ! Subroutines Follow
 920 !
 930 ! Find proper formulas
 940 N=0 @ FOR X=1 TO 4
 950 Z=2^(X-1) @ IF V(X) THEN N=N+Z          -Use array V() to create table
                                               to point to formula
 960 NEXT X @ N=IP(N/2)
 970 ON N GOSUB 1000,1020,1040,1060,1080
     ,1100
 980 RETURN                                  -Return from computation
                                               subroutine
 990 ! Given E & R                           -Six possible formula pairs
                                               follow
1000 V(3)=V(1)/V(2) @ V(4)=V(1)^2/V(2) @
      RETURN
1010 ! Given E & I
1020 V(2)=V(1)/V(3) @ V(4)=V(1)*V(3) @ R
     ETURN
1030 ! Given R & I
1040 V(1)=V(2)*V(3) @ V(4)=V(3)^2*V(2) @
      RETURN
1050 ! Given P & E
1060 V(2)=V(1)^2/V(4) @ V(3)=V(4)/V(1) @
      RETURN
1070 ! Given P & R
1080 V(1)=SQR(V(4)*V(2)) @ V(3)=SQR(V(4)
     /V(2)) @ RETURN
1090 ! Given P & I
1100 V(1)=V(4)/V(3) @ V(2)=V(4)/V(3)^2 @
      RETURN
1110 ! Display results
1120 FOR X=1 TO 4
1130 V=FNR(FNF(V(X)),2)                      -Round off and fix each answer
1140 DISP V;F$&A$[X*6-5,X*6]                 -Show value V and unit prefix
                                               and unit A$
1150 Q$=FNK$
1160 IF NUM(Q$)#8 AND NUM(Q$)#13 THEN 11     -Wait for RTN or BACK key
     50
1170 IF NUM(Q$)#8 THEN 1190
1180 X=X-2 @ IF X<0 THEN X=0                 -If BACK key adjust counter
1190 NEXT X
1200 GOSUB 1230                              -Prompt continuation options
                                               and return Q$
1210 IF Q$="V" THEN 1120                     -View answer again
1220 IF Q$="R" THEN 700 ELSE 1260            -Run again else stop program
1230 DISP CHR$(210);"un again, ";CHR$(21
     4);"iew again, or ";CHR$(197);
1240 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
1250 IF Q$="V" OR Q$="R" OR Q$="E" THEN      -Wait for R, V, or E before
     RETURN                                    returning
1260 DELAY 1 @ DISP @ STOP                   -Program always stops here.
```

PROGRAM DESCRIPTION

IMPEDANCE CONVERSION

Given impedance Z and Phase Angle Theta, the Resistance and Reactance, (R & X), Normalized Resistance and Reactance, R(N) & X(N), Reflection Coefficient and Associated Angle, VSWR and Return Loss can all be computed from the following formulas:

$$R = Z * \text{Cos}\ (\theta) \qquad X = Z * \text{SIN}\ (\theta)$$

$$R(n) = R/\text{characteristic } Z \qquad X(n) = X/\text{characteristic } Z$$

$$\text{Reflected Angle} = \frac{\sqrt{(R(n)-1)^2 + X(n)^2}}{\sqrt{(R(n)+1)^2 + X(n)^2}}$$

Reflection Coefficient = (57.2958 * ATN (X(n)/R(n))-1)-(57.2958 * ATN (X(n)/R(n))+1)

VSWR = (Reflected Angle + 1) / (1 - Reflected Angle)

Return Loss = -20 * .434294 * LOG (Reflected Angle)

SAMPLE PROBLEM

What is the VSWR and return loss on a 50Ω line with a characteristic impedance of 50Ω and a phase angle of +36.9 degrees?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	***IMPEDANCE CONVERSION***	
2	Enter impedance	Enter Impedance Z =	50 [RTN]
3	Enter phase angle	Enter angle =	36.9 [RTN]
4	Display R & X	R=39.985 X=30.021	[RTN]
5	Enter characteristic Z	Characteristic impedance =	50 [RTN]
6	Display R(n) & X(n)	R(n)=.8 X(n)=.6	[RTN]
7	Reflected Coefficient & theta	Ref Cof=.334 angle=89.999	[RTN]
8	Display VSWR & R.L.	VSWR=2.001 R.L.=9.535	[RTN]
9	End program	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	*** IMPEDANCE CONVERSION ***	
2	Enter impedance	Enter Impedance Z =	Z [RTN]
3	Enter phase angle	Enter angle =	Angle [RTN]
4	Display R & X	R=value X=value	[RTN]
5	Enter characteristic Z	Characteristic Impedance =	Impedance [RTN]
6	Display R(n) & X(n)	R(n)=value X(n)=value	[RTN]
7	Reflected coefficient & angle	Ref Coef=value Angle = value	[RTN]
8	Display VSWR & R.L.	VSWR=value R.L.=value	[RTN]
9	Program options	Run again, View again, or End? R	R, V, or E [RTN]
	If 'R' then goto 2		
	If 'V' then goto 6		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
I	Function variable	P3-P6	Intermediate values
J	Function variable	V4	Reflection coefficient
K	Formula constants	V5	Angle
L	Formula constants	Y3	VSWR
M	Formula constants	Y4	Return Loss
P1	R(n)	K$	Single key input
P2	X(n)	Q$	Continuation input

PROGRAM LISTING

```
 10 ! IMPEDANCE CONVERSIONS
 20 !
 30 ! Convert Z and Theta to
 40 ! R & X,     R(n) & X(n),
 50 !  and  Rho & VSWR.
 60 ! Revision 11/01/82
 70 !
 80 OPTION ANGLE RADIANS
 90 ! Define Functions
100 !
110 ! Single, upper case key in
120 DEF FNK$
130 K$=KEY$ @ IF K$="" THEN 130
140 FNK$=UPRC$(K$)
150 END DEF
160 !
170 ! Round 'I' to 'J' digits
180 DEF FNR(I,J)
190 I=INT(I*10^J+.5)/10^J
200 FNR=I
210 END DEF
220 !
230 ! Formula constants
240 K=.434294 @ L=57.2958 @ M=.017453
250 ! Sign on message
260 DISP "* * * IMPEDANCE CONVERSION *
    * *"
270 INPUT "Enter Impedance Z=  ";V1
280 INPUT "Enter angle =  ";V2
290 V2=V2*M                                -Adjust phase angle
300 Y1=FNR(V1*COS(V2),3) @ Y2=FNR(V1*SI    -Compute R and X
    N(V2),3)
310 DISP "R=";Y1;"    X=";Y2
320 K$=FNK$
330 INPUT "Characterastic Impedance = "
    ;V3
340 P1=Y1/V3 @ P2=Y2/V3                    -Compute R(N) and X(N)
350 DISP "R(n)=";FNR(P1,3);"     X(n)=";
    FNR(P2,3)
360 P3=SQR((P1+1)^2+P2^2)                  -Compute intermediate values
370 IF P1+1>0 THEN P4=ATN(P2/(P1+1)) EL
    SE P4=ATN(P2/(P1+1))-PI
380 P4=P4*L
390 P5=SQR((P1-1)^2+P2^2)
400 IF P1-1>0 THEN P6=ATN(P2/(P1-1)) EL
    SE P6=ATN(P2/(P1-1))-PI
410 P6=P6*L
420 V4=P5/P3 @ V5=P6-P4                    -Compute reflection coefficient
                                            and angle
430 IF V5<=180 AND V5>=-180 THEN 450       -Adjust angle if not in first
                                            or fourth quadrant
440 IF V5>180 THEN V5=V5-360 ELSE V5=V5    -Adjust angle
    +360
450 Y1=FNR(V4,3) @ Y2=FNR(V5,3)
```

PROGRAM LISTING

```
460 Y3=FNR((V4+1)/(1-V4),3)
470 Y4=FNR(-20*K*LOG(V4),3)            -Compute return loss
480 K$=FNK$
490 DISP "Ref Cof=";Y1;"  angle=";Y2
500 K$=FNK$
510 DISP "VSWR=";Y3;"   R.L.=";Y4
520 K$=FNK$
530 !
540 ! Continuation options
550 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
560 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
570 IF Q$#"V" AND Q$#"R" AND Q$#"E" THE
    N 550
580 IF Q$="V" THEN 350
590 IF Q$="R" THEN 270
600 STOP
```

PROGRAM DESCRIPTION

SMITH CHART CONVERSION

Rho, VSWR, and Return Loss are related by the following diagram.

$$Rho = 10^{(-R.L./20)} \qquad Rho = \frac{VSWR-1}{VSWR+1}$$



$$R.L. = ZO * LOG_{10}\ (1/Rho) \qquad VSWR = \frac{Rho-1}{Rho+1}$$

Given any value the other two are computed.

Practical operating range:

$0 < Rho < 1$

$0 < VSWR < ZO$

$1 < R\ L < ZO$

All three are displayed on a single display line.

SAMPLE PROBLEM

If the measured VSWR is 1.6, what are Rho and the return loss?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	* SMITH CHART CONVERSION *	
2	Skip Rho	Rho =	[RTN]
3	Skip R L	Return Loss =	[RTN]
4	Enter VSWR	VSWR =	1.6 [RTN]
5	Display results	Rho-.23 R L=12.74 VSWR=1.6	[RTN]
6	End program	Run again or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	* SMITH CHART CONVERSION *	
2		Rho =	Value [RTN] or [RTN]
3	If Rho was entered then 9		
4	Else	Return loss =	Value [RTN] or [RTN]
5	If R L was entered then 9		
6	Else	VSWR =	Value [RTN] or [RTN]
7	If VSWR was entered then 9		
8	Else goto 12		
9	Compute results		
10	Display results	Rho=R R L =L VSWR=V	[RTN]
11	Program options	Run again or End? R	E [RTN] or [RTN]
	If [RTN] then 2		
	If 'E' end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A	Temporary value	V	VWSR
B	Temporary value	K$	Single key input
F	Variable passed to function	Q$	Response to program options
J	Variable passed to function	X$	String input always converted to A by A=VAL(=0$)
L	Return loss (R.L.)		
R	Rho	P$	Prompt string passed to non-error input subroutine

NOTES AND REFERENCES

Practical limits: 0 Rho 1

0 VSWR Z0

1 R.L. Z0

Reference: From Hewlett-Packard Users' Library program number 00031D

PROGRAM LISTING

```
 10 ! SMITH CHART CONVERSION
 20 ! VSWR, Rho and Return Loss
 30 ! are each related and given
 40 ! any one the other two will
 50 ! be computed.
 60 ! Revision 11/01/82
 70 !
 80 DELAY 2
 90 DIM P$[20]
100 ! Define Functions
110 !
120 ! Single, upper case key in
130 DEF FNK$
140 K$=KEY$ @ IF K$="" THEN 140
150 FNK$=UPRC$(K$)
160 END DEF
170 !
180 ! Round 'F' to 'J' digits
190 DEF FNR(F,J)
200 F=INT(F*10^J+.5)/10^J
210 FNR=F
220 END DEF
230 !
240 ! Main program loop
245 DISP "   * SMITH CHART CONVERSION *
    "
250 R,V,L=0
260 P$="Rho = " @ GOSUB 550
270 IF A THEN 330                      -If Rho input≠0 then compute RL
                                        and VSWR
280 P$="Return Loss = " @ GOSUB 550
290 IF A THEN 380                      -If RL≠0 then compute Rho and
                                        VSWR
300 P$="VSWR = " @ GOSUB 550
310 IF A THEN 430 ELSE 500             -If A then compute else goto
                                        continue options
320 ! Compute VSWR & R.L.
330 R=A
340 A=R+1 @ B=R-1 @ V=-A/B
350 L=20*LOG10(1/R)
360 GOTO 470                           -Goto display
370 ! Compute Rho & VSWR
380 L=A
390 R=10^(-L/20)
400 A=R+1 @ B=R-1 @ V=-A/B
410 GOTO 470                           -Goto display
420 ! Compute Rho
430 V=A
440 A=V-1 @ B=V+1 @ R=A/B
450 GOTO 350                           -Go compute RL
460 ! Display results
470 DISP "Rho=";FNR(R,2);" RL=";FNR(L,2
    );" VSWR=";FNR(V,1)
480 K$=FNK$
```

PROGRAM LISTING

```
490 ! Continuation options
500 DISP CHR$(210);"un again or ";CHR$(
    197);
510 INPUT "nd  ","R"; Q$ @ Q$=UPRC$(Q$)
520 IF Q$#"R" AND Q$#"E" THEN 500
530 IF Q$="R" THEN 250 ELSE 600
540 ! Non-error input
550 DISP P$;
560 INPUT ""; X$ @ IF X$="" THEN X$="0"     -Return A=0 for null input
570 ON ERROR GOTO 550
580 A=VAL(X$) @ OFF ERROR
590 RETURN
600 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

MISMATCH

Mismatch uncertainty in dB can be computed given two VSWR's and the constant K=.434294 using the following formulas:

```
Rho1 = P1 =   (VSWR1-1) / (VSWR1+1)

Rho2 = P2 =   (VSWR2-1) / (VSWR2+1)

Positive error  =  20*K*LOG(1+P1*P2)

Negative error  =  20*K*LOG(1-P1*P2)
```

SAMPLE PROBLEM

What is the mismatch uncertainty in dB of two VSWR's V1=2.43 and V2=4.56?

Answer: +2.055 dB
-2.697 dB

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	***MISMATCH***	
2	Enter first VSWR	Enter first VSWR?	2.43 [RTN]
3	Enter second VSWR	Enter second VSWR?	4.56 [RTN]
4	Display positive error	Plus 2.055 dB	[RTN]
5	Display negative error	Minus 2.697 dB	[RTN]
6	End program	Run again, View again or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	*** MISMATCH ***	
2	Enter first VSWR	Enter first VSWR	First VSWR [RTN]
3	Enter second VSWR	Enter second VSWR	Second VSWR [RTN]
4	Display positive error	Plus -value- dB	[RTN]
5	Display negative error	Minus -value- dB	[RTN]
6	Program options	Run again, View again, or End? R	R, V, or E [RTN]
	If 'R' then step 2		
	If 'V' then step 4		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
I	Function variable	V1	VSWR 1
J	Function variable	V2	VSWR 2
K	Formula constant (.434294)	Y1	Positive error
P1	Rho of VSWR 1	Y2	Negative error
P2	Rho of VSWR 2	K$	Single key input
		Q$	Continuation option input

NOTES AND REFERENCES

Accuracy for VSWR >15 uncertain. From Series 80 Electrical Engineering Solution Book.

PROGRAM LISTING

```
 10 ! MISMATCH
 20 !
 30 ! Compute the mismatch
 40 ! uncertainty in dB based
 50 ! on two VSWR's.
 60 ! Revision 11/01/82
 70 !
 80 DELAY 2
 90 ! Define Functions
100 !
110 ! Single, upper case key in
120 DEF FNK$
130 K$=KEY$ @ IF K$="" THEN 130
140 FNK$=UPRC$(K$)
150 END DEF
160 !
170 ! Round 'I' to 'J' digits
180 DEF FNR(I,J)
190 I=INT(I*10^J+.5)/10^J
200 FNR=I
210 END DEF
220 !
230 ! Formula constant
240 K=.434294
250 ! Sign on message
260 DISP "  * * *  M I S M A T C H  * *
     *"
270 INPUT "Enter first VSWR ? ";V1
280 P1=(V1-1)/(V1+1)
290 INPUT "Enter second VSWR ? ";V2
300 P2=(V2-1)/(V2+1)
310 Y1=ABS(FNR(20*K*LOG(1+P1*P2),3))     -Drop sign as answer is the
                                          proper sign
320 Y2=ABS(FNR(20*K*LOG(1-P1*P2),3))     -Drop sign as answer label is
                                          the proper sign
330 DISP "Plus  ";Y1;"dB"
340 K$=FNK$
350 DISP "Minus  ";Y2;"dB"
360 K$=FNK$
370 ! Re-run option select
380 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
390 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
400 IF Q$#"V" AND Q$#"R" AND Q$#"E" THE
    N 380
410 IF Q$="V" THEN 330
420 IF Q$="R" THEN 270
430 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

dB TO % TO dB CONVERSION

It is sometimes necessary to convert a voltage or power change expressed as a percentage to a change expressed as a dB ratio.

1. To convert dB to % use the following formulas:

   $P2 = 10\uparrow(D/10)$ $P1 = \sqrt{P2}$ $D = dB$

   Voltage % increase = 100*(P1-1)

   Voltage % decrease = 100*(P1-1)/P1

   Power % increase = 100*(P2-1)

   Power % decrease = (P2-1)/P2

2. To convert % to a dB ratio use the following P=% change

   Voltage dB increase = $V1 = |(20*LOG_{10}\ (1/(1+P/100))|$

   Voltage dB decrease = $V2 = |(20*LOG_{10}\ (1/(1-P/100))|$

   Power dB increase = P1 = V1/2

   Power dB decrease = P2 = V2/2

Limited range is: 0<dB <30

For % a value > = 100 will not compute a decrease as decreases can not exceed 100%.

SAMPLE PROBLEM

A) What is the % of power increase for a dB increase of 4.8?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
A 1	Run program	** dB to % to dB Conversion **	
2	Select operation D	Select: % >dB or dB >%? %	D [RTN]
3	Enter dB ratio	Enter: dB =	4.8 [RTN]
4	Display voltage %	Voltage = +73.78% -42.46%	[RTN]
5	Display power %	Power = +202% -66.89%	[RTN]
6	End program	Run again, View again, or End? R	E [RTN]

SAMPLE PROBLEM

B) What is the dB change for a 45% change in voltage?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
B_1	Run program	** dB to % to dB Conversion **	
2	Select default %	Select % >dB or dB >%? %	[RTN]
3	Enter % of change	Enter: Percent =	45 [RTN]
4	Display voltage, dB	Voltage = +3.23dB -5.19dB	[RTN]
5	Display power, dB	Power = +1.61dB -2.59dB	[RTN]
6	End program	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	** dB to % to dB Conversion **	
2	Select operation	Select: % >dB or dB >%? %	[RTN] or D [RTN]
3	If D then goto 8		
4	Enter % of change	Enter: Percent =	%value [RTN]
5	Display voltage dB	Voltage = +Val dB -Val dB	[RTN]
6	Display power in dB	Power = +Val dB -Val dB	[RTN]
7	Program options:	Run again, View again, or End? R	R, V, or E [RTN]
	If 'R' then 4		
	If 'V' then 5		
	If 'E' then end program		
8	Enter dB ratio	Enter: dB =	dB value [RTN]
9	Display voltage %	Voltage = +Val % -Val %	[RTN]
10	Display power %	Power = +Val % -Val %	[RTN]
11	Program options	Run again, View again, or End? R	R, V, or E [RTN]
	If 'R' then 4		
	If 'V' then 5		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
D	dB input	V3	Power dB increase
I	Function variable	V4	Power dB decrease
J	Function variable	V7	Voltage intermediate variable
P1	Intermediate voltage	Y1	Voltage % increase
P2	Intermediate power	Y2	Voltage % decrease
Q	Flag 1=%>dB 0=dB>%	Y3	Power % increase
V1	Voltage dB increase	Y4	Power % decrease
V2	Voltage dB decrease	K$	Single key input
		Q$	Continuation options input

PROGRAM LISTING

```
 10 ! dB to % to dB CONVERSION
 20 ! Convert a dB ratio to a
 30 ! % of voltage and current
 40 ! increase and decrease and
 50 ! visa-versa.
 60 ! Revision 11/01/82
 70 !
 80 DELAY 2
 90 ! Define Functions
100 !
110 ! Single, upper case key in
120 DEF FNK$
130 K$=KEY$ @ IF K$="" THEN 130
140 FNK$=UPRC$(K$)
150 END DEF
160 !
170 ! Round 'I' to 'J' digits
180 DEF FNR(I,J)
190 I=INT(I*10^J+.5)/10^J
200 FNR=I
210 END DEF
220 !
230 ! Sign on message
240 DISP "* * dB to % to dB Conversion
    * *"
250 DISP "Select:  ";CHR$(165);" > dB
    or  ";CHR$(228);"B > % ";
260 INPUT "?","%"; Q$ @ Q$=UPRC$(Q$)
270 IF Q$#"%" AND Q$#"D" THEN 250          -Wait for proper selection,
                                             either '%' or 'D'
280 IF Q$="%" THEN 450
290 ! Convert dB to percent
300 Q=0 @ INPUT "Enter:  dB = ";D
310 D=ABS(D) @ IF D>30 THEN 300            -Keep dB positive and <=30 else
                                             answer too long
320 P2=10^(D/10)
330 P1=SQR(P2)
340 ! Convert + & - Voltages
350 V7=P1 @ GOSUB 430
360 Y1=FNR(V3,2) @ Y2=FNR(V4,2)
370 ! Convert + & - Powers
380 V7=P2 @ GOSUB 430
390 Y3=FNR(V3,2) @ Y4=FNR(V4,2)
400 DISP "Voltage =   +";Y1;"%  ";Y2;"%  -Display voltage as % and wait
    " @ K$=FNK$                              for key
410 DISP "Power =  +";Y3;"%  ";Y4;"%" @   -Display power as % and wait
     K$=FNK$                                 for key
420 GOTO 530                               -Goto continuation options
430 V3=100*(V7-1) @ V4=-100*(V7-1)/V7 @   -Subroutine for dB to %
     RETURN
440 ! Convert percent to dB
450 Q=1 @ INPUT "Enter:  Percent = ";P
460 V1=ABS(20*LOG10(1/(1+P/100)))
470 IF P>=100 THEN P=0
```

PROGRAM LISTING

```
480 V2=ABS(20*LOG10(1/(1-P/100)))
490 V3=FNR(V1/2,2) @ V4=FNR(V2/2,3) @ V
    1=FNR(V1,2) @ V2=FNR(V2,2)
500 DISP "Voltage =  +";V1;"db   -";V2;   -Display voltage in dB and wait
    "dB" @ K$=FNK$                            for key
510 DISP "Power =   +";V3;"dB   -";V4;"   -Display power in dB and wait
    dB" @ K$=FNK$                             for key
520 ! Continuation options
530 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
540 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
550 IF Q$#"V" AND Q$#"R" AND Q$#"E" THE
    N 530
560 IF Q$#"V" THEN 580
570 IF Q THEN 500 ELSE 400                 -Goto proper view routine based
                                            on Q
580 IF Q$#"R" THEN 600
590 IF Q THEN 450 ELSE 300                 -Goto proper run routine based
                                            on Q
600 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

BUTTERWORTH FILTER DESIGN

The common Butterworth Filter (BF) is made up of an odd number of components with each section of a low pass filter looking like this:



And each section of a High Pass filter like this:



Where 'T' is the final terminating element.

L_K & C_K coefficients are determined by:

$Q_K = Z * \sin (2*K-1) * P1/2*N$ where K is each element number 1 thru N and N is total or last element number.

Actual component values are determined by:

$C_K = Q_K / (R*W)$ $\quad$ $L_K = Q_K * R/W$ for low pass.

$C_K = 1/(Q_K *R*W)$ $\quad$ $L_K = R*(1/Q_K /W)$

In this program $3 \leq N \leq 19$ and is always ODD.

To use the program provide the number of elements, cut-off frequency, and terminating impedance. Choose between a low pass or high pass filter. The program will display values for all the C_{in} sections followed by the single termination value. The program will then display the corresponding values for L_{in}, and the termination value.

The display can be stepped forward with the [RTN] key or backward with the [BACK] key.

Negative values on any component means the filter cannot be designed properly around the given parameters.

SAMPLE PROBLEM

Design a Low Pass filter with a cut-off frequency of 10 MHz and five elements as shown in the schematic. Assume a 50Ω terminating impedance.



SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	*BUTTERWORTH FILTER DESIGN*	
2	Set number of element	Number of elements (<=19)?	5 [RTN]
3	Select Low Pass	Low Pass or High Pass?	L [RTN]
4	Set cut-off frequency	Cut-off frequency?	10000000 [RTN]
5	Set terminating impedance	Terminating impedance?	50 [RTN]
6		Low pass filter [RTN] to VIEW	[RTN]
7	Display values:	C(in) 1 196.73pfd 1.29μH	[RTN]
8		C(in) 2 636.62pfd 1.29μH	[RTN]
9		C(in) 3 196.726pfd	[RTN]
10		L(in) 1 491.82nH 515.04pfd	[RTN]
11		L(in) 2 1.59μH 515.04pfd	[RTN]
12		L(in) 3 491.816nH	[RTN]
13	End program	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	*BUTTERWORTH FILTER DESIGN*	
2	Enter number of elements	Number of elements (<=19)?	N [RTN]
3	Enter Low or High Pass	Low Pass or High Pass?	L or H [RTN]
4	Enter cut-off frequency	Cut-off frequency?	W [RTN]
5	Enter terminating impedance	Terminating impedance?	R [RTN]
6	Display output section	High pass filter..[RTN] to view	[RTN]
7	or	Low pass filter..[RTN] to view	[RTN]
8	First C input element	C(in)1 C1 L1	[RTN]/[BACK]
9	Continue display on		[RTN]
10	Termination element	C(in)n C_n	
11	First L input element	L(in)1 L1 C1	[RTN]/[BACK]
12	Continue display on		[RTN]
13	Terminating L element	L(in)n L_n	[RTN]
14	Program options	Run again, View again, or End? R	R,V, or E [RTN]
15	If 'V' goto 6		
16	If 'R' goto 2		
17	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
C(n)	Capacitor array	W	Cut-off frequency
F	Value passed to functions	Z	ON Z GOTO pointer
G(n)	Coefficient array	Q	Section counter
I	Loop counter	F$	Returned from function
J	Value passed to function	K$	Single key input
L(n)	Inductance array	O$	String inputs
R	Termination impedance	M$	Display header string
N	Number of elements		

NOTES AND REFERENCES

NOTE: Any negative computed value indicates the filter can not be designed properly with the parameters given.

From Series 80 Electrical Engineering Solution Book.

PROGRAM LISTING

```
10 ! PASSIVE BUTTERWORTH
20 ! FILTER DESIGN
30 ! Compute the inductor and
40 ! capicator combinations,
50 ! required to construct a
60 ! passive butterworth filter.
70 ! Revision 11/01/82
80 DELAY 2
90 ! Define Functions
100 !
110 ! Wait for a 'RTN'
120 DEF FNK$
130 K$=KEY$ @ IF K$="" THEN 130
140 IF NUM(K$)#13 THEN 130
150 FNK$=K$
160 END DEF
170 !
180 ! Round 'F' to 'J' digits
190 DEF FNR(F,J)
200 F=INT(F*10^J+.5)/10^J
210 FNR=F
220 END DEF
230 !
240 ! Fix values to common units.
250 DEF FNF(F)
260 J=3 @ F$="MK m"&CHR$(12)&"np"              -CHR$(12) is micro symbol
270 IF F<1000 AND F>=1 THEN 320
280 IF F>1000 THEN 300
290 F=F*1000 @ J=J+1 @ GOTO 310
300 F=F/1000 @ J=J-1
310 IF J#1 AND J#7 THEN 270
320 IF J=3 THEN F$="" ELSE F$=F$[J,J]
330 FNF=F
340 END DEF
350 !
360 DIM C(20),G(20),L(20)                     -G is coefficient of Ck and Lk
370 ! Main program loop
380 DISP " *  BUTTERWORTH FILTER DESIGN
   *"
390 DELAY 0
400 INPUT "Number of elements ( <=19 )?       -Be sure that N is positive
   "; N @ N=ABS(N)
410 IF N/2=INT(N/2) OR N>19 OR N<3 THEN      -N must be odd
    400
420 FOR I=1 TO N
430 G(I)=2*SIN((2*I-1)*PI/(2*N))             -Set coefficient for each
                                              element
440 NEXT I
450 DISP CHR$(204);"ow pass or ";CHR$(2
   00);"igh pass ";
460 INPUT "?"; Q$ @ Q$=UPRC$(Q$)
470 IF Q$#"H" AND Q$#"L" THEN 450             -Wait for 'H' or 'L'
480 IF Q$="L" THEN Z=1                        -Z is flag for High Pass or Low
                                              Pass
```

PROGRAM LISTING

```
490 IF Q$="H" THEN Z=2
500 INPUT "Cut-off frequency?";W
510 W=2*PI*W
520 INPUT "Terminating impedance?";R
530 ON Z GOTO 550,610                      -Select filter type
540 ! Low pass
550 FOR I=1 TO N
560 C(I)=G(I)/(R*W) @ L(I)=G(I)*R/W
570 NEXT I
580 M$="Low pass filter..."                -Display header message
590 GOTO 660
600 ! High pass
610 FOR I=1 TO N
620 C(I)=1/(G(I)*R*W) @ L(I)=R*(1/G(I)/
    W)
630 NEXT I
640 M$="High pass filter.."                -Display header message
650 ! Display results
660 DISP M$;
670 INPUT "[RTN] to view";Q$
680 FOR I=1 TO N-2 STEP 2                  -Loop to display C(input)
                                            values
690 Q=IP(I/2)+1                            -Q is the section counter
700 DISP "C(in)";Q;
710 DISP FNR(FNF(C(I)),2);F$&"fd  ";
720 DISP FNR(FNF(L(I+1)),2);F$&"H"
730 K$=KEY$ @ IF K$="" THEN 730
740 IF NUM(K$)#8 AND NUM(K$)#13 THEN 73
    0
750 IF NUM(K$)=13 THEN 770
760 I=I-4 @ IF I<0 THEN I=-1               -Step back to previous values
770 NEXT I
780 DISP "C(in)";Q+1;"   ";FNR(FNF(C(N)
    ),3);F$&"fd"
790 K$=FNK$
800 FOR I=1 TO N-2 STEP 2                  -Loop to display L(input)
                                            values
810 Q=IP(I/2)+1
820 DISP "L(in)";Q;
830 DISP FNR(FNF(L(I)),2);F$&"H  ";
840 DISP FNR(FNF(C(I+1)),2);F$&"fd"
850 K$=KEY$ @ IF K$="" THEN 850
860 IF NUM(K$)#13 AND NUM(K$)#8 THEN 85
    0
870 IF NUM(K$)=13 THEN 890
880 I=I-4 @ IF I<0 THEN I=-1               -Step back to previous answer
890 NEXT I
900 DISP "L(in)";Q+1;"   ";FNR(FNF(L(N)
    ),3);F$&"H"
910 K$=FNK$
920 ! Continuation options
930 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
940 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
950 ON POS('RVE',Q$)+1 GOTO 930,400,650
    ,960
960 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

ACTIVE FILTER DESIGN

Using constants from a published data table, resistors for high pass filters or capacitors for low pass filters can be computed using the formulas:

High Pass	RA=K/(12.56 * F * C) * 2	&	RB=1/(RA * C^2*(2PI * F)^2)
Low Pass	CA=1/(R * (2PI * F) * K)	&	CB=K/(2PI * F * R)
	K=constant		

In this program, limited to four stages or less, K is a selected set of values based on the number of stages.

Single stage	K=K(1)	
Two stages	K=K(2)-K(3)	Array K is loaded with the proper
Three stages	K=K(4)-K(6)	values one time early in program
Four stages	K=K(7)-K(10)	initialization.

Results are computed into array A(x) and displayed two at a time for each stage.

SAMPLE PROBLEM

Using the schematic, design a high pass filter with a cut-off frequency of 1KHz, attenuation of 48 db per octave and a capacitor value of .001 mfd.



To cascade stages, change the component labels to their label plus the stage number. i.e.: for three stages:

Stage 1	R1	R2
	C1	C2
Stage 2	R3	R4
	C3	C4
Stage 3	R5	R6
	C5	C6

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	** ACTIVE FILTER DESIGN **	
2	Enter cut-off frequency	Cut-off frequency in Hz?	1000 [RTN]
3	Select filter mode	High pass or Low pass? L	H [RTN]
4	Select att. per octave	Attneuation (12,24,36,48)? 12	48 [RTN]
5	Enter value for C	C in microfarads =	.001 [RTN]
6	Display section values for	S1: R1=31.1KΩ R2=816.2KΩ	[RTN]
	each section S	S2: R3=88.5KΩ R4=286.6KΩ	[RTN]
		S3: R5=132.4KΩ R6=191.5KΩ	[RTN]
		S4: R7=156.2KΩ R8=162.4KΩ	[RTN]
7	End program	Run again, View again, or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	** ACTIVE FILTER DESIGN**	
2	Enter cut-off frequency	Cut-off frequency in Hz?	F [RTN]
3	Enter High or Low Pass	High Pass or Low Pass? L	H [RTN] or [RTN]
4	Enter attenuation per octave	Attenuation (12,24,36,48)? 12	Select [RTN]
5	If Low Pass	R in Ohms =	R [RTN]
6	If High Pass	C in microfarads =	C [RTN]
7	Display results:		
8	If Low Pass	S#: C#=value C#+1=value	[RTN]
9	If High Pass	S#: R#=value R#+1=value	[RTN]
10	Repeat for number of sections		
	based on attenuation per		
	octave		
11	Program options	Run again, View again, or End? R	R, V, or E [RTN]
	If 'R' goto 2		
	If 'V' goto 7		
	If 'E' end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A	Attenuation token	Z	Constant K(z) pointer
F	Cut-off frequency	A( )	Answer array
I	Function variable	K( )	Constant array
J	Function variable	F$	Function string
P	High Pass=1 Low Pass=0 Flag	K$	Single key input
S	Section number for display	L$	Label string
V	Capacitor Value on High Pass, Resistor Value on Low Pass	P$	Prompt string
		U$	Units string
X	Loop counter	Q$	Options input

NOTES AND REFERENCES

When viewing results, step forward with the [RTN] key and backward with the [BACK] key.

Answers greater than 999M will "spill" off the display.

PROGRAM LISTING

```
10 ! ACTIVE FILTER DESIGN
20 ! Computes resistor and
30 ! capacitor combinations
40 ! needed to design an
50 ! active, OP AMP filter.
60 ! Revision 11/01/82
70 !
80 DELAY 2
90 DIM A(8),K(10),P$[20],L$[5],U$[2],Q
   $[1]
100 FOR X=1 TO 10 @ READ K(X) @ NEXT X      -Load constant array K(X) with
                                              vales from DATA stmt
110 DATA .707,.3827,.9239,.2588,.7071,.
    9659,.1951,.5556,.8315,.9808
120 ! Define Functions
130 ! Fix answer to common unit
140 DEF FNF(I)
150 F$="MK m"&CHR$(12)&"np"                 -CHR$(12) is symbol for micro
160 J=3
170 IF I<1000 AND I>=1 THEN 250
180 IF I<1 THEN 220
190 IF I=1 THEN 250
200 I=I/1000 @ J=J-1
210 GOTO 170
220 IF J=7 THEN 250
230 I=I*1000 @ J=J+1
240 GOTO 170
250 F$=F$[J,J]
260 FNF=I
270 END DEF
280 !
290 ! Single, upper case key in
300 DEF FNK$
310 K$=KEY$ @ IF K$="" THEN 310
320 FNK$=UPRC$(K$)
330 END DEF
340 !
350 ! Round 'F' to 'J' digits
360 DEF FNR(I,J)
370 I=INT(I*10^J+.5)/10^J
380 FNR=I
390 END DEF
400 !
410 ! Sign on message
420 DISP "  * * ACTIVE FILTER DESIGN *
    *"
430 INPUT "Cut-off frequency in Hz? ";F
440 DISP CHR$(200);"igh pass or ";CHR$(
    204);
450 INPUT "ow pass ?","L"; Q$ @ Q$=UPRC     -Select High-Pass or Low-Pass
    $(Q$)
460 IF Q$#"H" AND Q$#"L" THEN 440
470 ! Set Hi / Lo flag P
480 IF Q$="L" THEN P=0 ELSE P=1
```

PROGRAM LISTING

```
490 ! Set prompt label P$
500 IF P THEN P$="C in microfarads = "
    ELSE P$="R in ohms = "
510 ! Set answer label L$
520 IF P THEN L$="R" ELSE L$="C"
530 ! Set answer units U$
540 IF P THEN U$=CHR$(17) ELSE U$="fd"      -Set units string to ohms if
                                              high pass else 'fd'
550 INPUT "Attenuation (12, 24, 36, 48)
    ?","12";A
560 IF A/12#IP(A/12) THEN 550 ELSE A=A/     -Set attenuation level
    12
570 IF A<1 OR A>4 THEN 550
580 DISP P$; @ INPUT "";V
590 IF P THEN V=V*.000001                   -If High-Pass adjust to farads
600 Z=0 @ FOR X=1 TO A @ Z=X+Z-1 @ NEXT
     X @ Z=Z+1
610 FOR X=1 TO A*2 STEP 2
620 IF P THEN 650                           -Select formula
630 A(X)=1/(V*(6.28*F)*K(Z))
640 A(X+1)=K(Z)/(6.28*F*V) @ GOTO 670
650 A(X)=K(Z)/(12.56*F*V)*2
660 A(X+1)=1/(A(X)*V^2*(6.28*F)^2)
670 Z=Z+1                                   -Increment constant pointer
680 NEXT X
690 ! Display the answers
700 S=1                                     -S = section pointer
710 FOR X=1 TO A*2 STEP 2
720 DISP "S"&STR$(S)&": ";
730 DISP L$&STR$(X)&"=";FNR(FNF(A(X)),1
    );F$&U$&"    ";
740 DISP L$&STR$(X+1)&"=";FNR(FNF(A(X+1
    )),1);F$&U$
750 K$=FNK$                                 -Wait for keyboard entry
760 IF NUM(K$)#13 AND NUM(K$)#8 THEN 75
    0
770 IF NUM(K$)=13 THEN 810
780 ! Step backwards
790 X=X-4 @ IF X<0 THEN X=-1                -Decrement loop and section
                                              number
800 S=S-2 @ IF S<1 THEN S=0
810 S=S+1 @ NEXT X                          -Increment section number and
                                              loop counter
820 ! Re-run option select
830 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
840 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
850 IF Q$#"V" AND Q$#"R" AND Q$#"E" THE
    N 830
860 IF Q$="V" THEN 700
870 IF Q$="R" THEN 430
880 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

LOW PASS FILTER

Component values in a single section Low Pass Filter without attenuators are related by the following formulas and schematic. Given cut-off frequency and impedance, the program calculates the component value.



F = Cut-off Frequency

R = Characteristic Impendance

LØ = R/PI * F

L = .64 * LØ/1.2

L1 = LØ/2 + L2

L2 = .6 * LØ/2

C = 1/(PI*F*R)

C1 = .6 * C/2

SAMPLE PROBLEM

What components are required to construct a low pass filter with a 1MHz cut-off frequency and a characteristic impedance of 50 ohms? (see schematic)

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	* LOW PASS FILTER DESIGN *	
2	Enter cut-off frequency	Cut-off Frequency in Hz?	1000000[RTN]
3	Enter characteristic impedance	Impendance in Ohms?	50 [RTN]
4	Display L and C1	L=8.49μH C1=1.91nfd	[RTN]
5	Display L1 and L2	L1=12.73μH L2=4.77μH	[RTN]
6	Display C	C=6.73nfd	[RTN]
7	End program	Run again, View again, or End?	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	* LOW PASS FILTER DESIGN *	
2	Enter cut-off frequency	Cut-off Frequency in Hz?	Freq [RTN]
3	Enter characteristic impedance	Impedance in Ohms?	Zc [RTN]
4	Display L and C1	L=value C1=value	[RTN]
5	Display L1 and L2	L1=value L2=value	[RTN]
6	Display C	C=value	[RTN]
7	Continuation options	Run again, View again, or End? R	R, V, or E [RTN]
	If 'R' goto 2		
	If 'V' goto 4		
	If 'E' end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
F	Function variable	C1	Circuit components See schematic
J	Function variable	R	Circuit components See schematic
L	Circuit components See schematic	F	Cut-off frequency
L1	Circuit components See schematic	K$	Single key input
L2	Circuit components See schematic	Q$	Option string inputs
C	Circuit components See schematic	F$	Function string

PROGRAM LISTING

```
 10 ! LOW PASS FILTER DESIGN
 20 ! Compute the inductor and
 30 ! capicator combinations,
 40 ! required to construct a
 50 ! a low pass filter.
 60 ! Revision 11/01/82
 70 DELAY 2
 80 ! Define Functions
 90 !
100 ! Wait for a 'RTN'
110 DEF FNK$
120 K$=KEY$ @ IF K$="" THEN 120
130 IF NUM(K$)#13 THEN 120
140 FNK$=K$
150 END DEF
160 !
170 ! Round 'F' to 'J' digits
180 DEF FNR(F,J)
190 F=INT(F*10^J+.5)/10^J
200 FNR=F
210 END DEF
220 !
230 ! Fix values to common units.
240 DEF FNF(F)
250 J=3 @ F$="MK m"&CHR$(12)&"np"
260 IF F<1000 AND F>=1 THEN 310
270 IF F>1000 THEN 290
280 F=F*1000 @ J=J+1 @ GOTO 300
290 F=F/1000 @ J=J-1
300 IF J#1 AND J#7 THEN 260
310 IF J=3 THEN F$="" ELSE F$=F$[J,J]
320 FNF=F
330 END DEF
340 !
350 ! Main program loop
360 DISP "    *  LOW PASS FILTER DESIGN
     *"
370 L,L1,L2,C,C1,R=0                        -Clear variables
380 DELAY 0
390 INPUT "Cut-off Frequency in Hz ? ";
    F
400 INPUT "Impedance in Ohms? ";R
410 ! Compute values
420 L0=R/(PI*F)
430 C=1/(PI*F*R)
440 L2=.6*L0/2
450 L=.64*L0/1.2
460 C1=.6*C/2
470 L1=L0/2+L2
480 ! Display values
490 DISP "L=";FNR(FNF(L),2);F$&"H     C1
    =";FNR(FNF(C1),3);F$&"fd"
500 K$=FNK$                                 -Wait for 'RTN' to continue
510 DISP "L1=";FNR(FNF(L1),2);F$&"H    L
    2=";FNR(FNF(L2),2);F$&"H"
```

PROGRAM LISTING

```
520 K$=FNK$                                    -Wait for 'RTN' to continue
                                                display

530 DISP "C=";FNR(FNF(C),2);F$&"fd"
540 ! Continuation options
550 K$=FNK$                                    -Wait for 'RTN' to continue
                                                display

560 DISP CHR$(210);"un again, ";CHR$(21
    4);"iew again, or ";CHR$(197);
570 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)
580 IF Q$="R" THEN 390
590 IF Q$="E" THEN 610
600 IF Q$="V" THEN 490 ELSE 560
610 DELAY 1 @ DISP @ STOP
```

PROGRAM DESCRIPTION

COIL DESIGN

The approximate inductance of a single-layer, air-core coil can be calculated from the simplified formula:

$$L\ (\mu H) = A^2 * N^2/(9*A+10*B)$$

Where:

L = Inductance in microhenrys

A = Coil radius in inches

B = Coil length in inches

N = Number of turns

Likewise, to approximate the number of turns of a single-layer, air-core coil for a specified inductance use the formula:

$$N = \sqrt{L*(9*A+10*B)/A^2}$$

Given A, B, and N the program calculates inductance (L).

Given A, B, and L the program calculates the number of turns (N).

SAMPLE PROBLEM

What is the approximate inductance of a single-layer, air-core coil with a diameter 3/4", a length of 1½" and 48 turns?

SOLUTION

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	** COIL DESIGN **	
2	Select inductance	Find: Turns or Inductance? I	[RTN]
3	Enter diameter	Coil diameter in inches =	.75 [RTN]
4	Enter length	Coil length in inches =	1.5 [RTN]
5	Enter number of turns	Total turns =	48 [RTN]
6	Display results	Inductance is 17.6 µH	[RTN]
7	End program	Run again or End? R	E [RTN]

USER INSTRUCTIONS

STEP	INSTRUCTIONS	DISPLAY	INPUT
1	Run program	** COIL DESIGN **	
2	Select option	Find: Turns or Inductance? I	T [RTN] or [RTN]
3	If 'I' then goto 8 else		
	enter inductance	Inductance in µH =	Inductance [RTN]
4	Enter coil diameter	Coil diameter =	Diameter [RTN]
5	Enter coil length	Coil length =	Length [RTN]
6	Display results	N turns	[RTN]
7	Goto 12		
8	Enter coil diameter	Coil diameter in inches =	Diameter [RTN]
9	Enter coil length	Coil length in inches =	Length [RTN]
10	Enter turns	Total turns =	Turns [RTN]
11	Display results	Inductance is L µH	[RTN]
12	Program options	Run again or End? R	E [RTN] or [RTN]
	If [RTN] goto 3 or 8		
	If 'E' then end program		

VARIABLE NAMES

NAME	DESCRIPTION	NAME	DESCRIPTION
A	Coil diameter	I	Function variable
B	Coil length	J	Function variable
L	Inductance in µH	K$	Single key input
N	Number of turns in coil	Q$	Continuation option input
Q	Flag Q=1=Turns Q=0=Inductance		

NOTES AND REFERENCES

This is a simple approximation program for common values of inductance found in Amateur Radio Projects. Practical limit is less than 100 µH.

Formulas are from: Robert Myers (editor)
The RADIO AMATEUR'S HANDBOOK
American Radio Relay League, 1975, Page 26

PROGRAM LISTING

```
 10 ! COIL DESIGN
 20 ! Compute the inductance of
 30 ! a coil given diameter,
 40 ! number of turns,
 50 ! and length OR compute the
 60 ! number of turns given the
 70 ! inductance, diameter and
 80 ! length of a coil.
 90 ! Revision 11/01/82
100 !
110 DELAY 2
120 ! Define Functions
130 !
140 ! Single, upper case key in
150 DEF FNK$
160 K$=KEY$ @ IF K$="" THEN 160
170 FNK$=UPRC$(K$)
180 END DEF
190 !
200 ! Round 'I' to 'J' digits
210 DEF FNR(I,J)
220 I=INT(I*10^J+.5)/10^J
230 FNR=I
240 END DEF
250 !
260 ! Sign on message
270 DISP "  * * C O I L   D E S I G N *
    *"
280 DISP "Find:    ";CHR$(212);"urns or
    ";CHR$(201);"nductance ";
290 INPUT "?","I"; Q$ @ Q$=UPRC$(Q$)          -Select operation
300 IF Q$="I" THEN 410                        -If Q$='I' compute inductance
310 IF Q$#"T" THEN 280
320 ! Compute number of turns
330 Q=1 @ DISP "Inductance in ";CHR$(12
    );"H = ";
340 INPUT "";L
350 INPUT "Coil diameter in inches = ";
     A @ A=A/2
360 INPUT "Coil length in inches = ";B
370 N=FNR(SQR(L*(9*A+10*B)/A^2),1)           -Formula to compute number of
                                               turns
380 DISP N;"turns" @ K$=FNK$
390 GOTO 480                                  -Goto continuation options
                                               after viewing answer
400 ! Compute inductance
410 Q=0 @ INPUT "Coil diameter in inche
    s = "; A @ A=A/2
420 INPUT "Coil length in inches = ";B
430 INPUT "Total turns = ";N
440 L=A^2*N^2/(9*A+10*B)                     -Formula to compute inductace
450 DISP "Inductance is ";FNR(L,1);" "&
    CHR$(12)&"H"
460 K$=FNK$
```

PROGRAM LISTING

```
470 ! Continue option select
480 DISP CHR$(210);"un again or ";CHR$(
    197);
490 INPUT "nd?","R"; Q$ @ Q$=UPRC$(Q$)      -Select continuation option
500 IF Q$≠"R" AND Q$≠"E" THEN 480
510 IF Q$≠"R" THEN 530
520 IF Q THEN 330 ELSE 410                  -If Q=1 then compute turns else
                                             compute inductance
530 DELAY 1 @ DISP @ STOP
```

NOTES

NOTES

NOTES

NOTES

NOTES

ELECTRONICS

HEWLETT PACKARD

00075-90066 Printed in U.S.A.

Reorder No.
00075-13008